```
(* Mathematica is a higher level language in which one
  can perform symbol manipulation and substitution. As a result,
 we can essentially input the definition of the bracket
  polynomial and Mathematica automatically calculates it.
   This notebook illustrates how to calculate the three variable bracket
  polynomial RawBracket[K] and the bracket polynomial B[K] for the trefoil knot,
 the figure eight knot, the Hopf link, the rational knot Num[3,1,1,3],
 the knot N36 = 9_36 and the knot N42 =9_42. These last two examples
    refer to the labeling of these knots in the knot tables. N42 is the
    first example of a knot that is not equivalent to its mirror image,
 but undetectable from this mirror image by the normalized bracket
  polynomial. The last example in the notebook is a link of two components,
 discovered in January 2000 by Morwen Thistlethwaite that has the same bracket
   polynomial (hence same Jones polynomial) as an unlink of two components. At this
   date there is no example of a knot (one component) that cannot be detected by the
   Jones polynomial. The Mathematica program used herein is due to Dror Bar-Natan. *)
```

```
Trefoil = X[a, d, b, e] X[e, b, f, c] X[c, f, d, a]
```

```
X[a, d, b, e] X[c, f, d, a] X[e, b, f, c]
```

```
(* Label each edge of the shadow graph of the knot.Then list X[i,j,k,l] where
  i,j,k,l is the counter-
  clockwise order of the edge labels as seen starting from an overcrossing line. *)
(* This means that a single cyclic permutation of the crossing
  code gives the code for the switched crossing. *)
```

```
rule1 = {X[a_, b_, c_, d_] :> A del[a d] del[b c] +
                B del[a b] del[c d]};

rule2 = {del[a_ b_] del[b_ c_] :> del[a c]};
rule3 = {(del[_])^2 :> J, del[_^2] :> J};

RawBracket[t_] := Simplify[(t /. rule1 // Expand) //. rule2 /. rule3]
```

```
(* These rules encode the expansion formula for the bracket state sum. *)
```

```
RawBracket[X[a, b, c, d]]
```

A del[b c] del[a d] + B del[a b] del[c d]

```
TRAW = RawBracket[Trefoil] / J
```

$3 A^2 B + A^3 J + 3 A B^2 J + B^3 J^2$

```
Eight = X[e, b, d, a] X[a, f, h, e] X[f, d, g, c] X[b, h, c, g]
RawBracket[Eight] / J
```

General::spell1 : Possible spelling error: new symbol name "Eight" is similar to existing symbol "Right".

```
X[a, f, h, e] X[b, h, c, g] X[e, b, d, a] X[f, d, g, c]
```

$4 A^3 B J + 4 A B^3 J + A^4 J^2 + B^4 J^2 + A^2 B^2 (5 + J^2)$

```
rule4 = {B :> 1 / A, J :> -A^2 - 1 / A^2}
```

$\left\{B :\to \frac{1}{A},\ J :\to -A^2 - \frac{1}{A^2}\right\}$

```
B[t_] := Simplify[RawBracket[t] / J /. rule4]
T[t_] :=  Simplify[RawBracket[t] /. rule4]

SUB[t_] := Simplify[t /. rule4]

B[Trefoil]
SUB[TRAW]
```

$-\frac{-1 + A^4 + A^{12}}{A^7}$

$-\frac{-1 + A^4 + A^{12}}{A^7}$

$\mathbf{T = -\frac{-1 + A^4 + A^{12}}{A^7}}$

$-\frac{-1 + A^4 + A^{12}}{A^7}$

```
(* Recording value of trefoil as T. *)

RawBracket[Eight] / J
B[Eight]
```

$4\,A^3\,B\,J + 4\,A\,B^3\,J + A^4\,J^2 + B^4\,J^2 + A^2\,B^2\,(5 + J^2)$

$1 + \frac{1}{A^8} - \frac{1}{A^4} - A^4 + A^8$

$\mathbf{EG = 1 + \frac{1}{A^8} - \frac{1}{A^4} - A^4 + A^8}$

$1 + \frac{1}{A^8} - \frac{1}{A^4} - A^4 + A^8$

```
(* Recording value of Figure Eight Knot as EG. *)
```

```
Hopf = X[b, B, a, A] X[B, b, A, a]
RawBracket[Hopf] / J
B[Hopf]
```

X[b, B, a, A] X[B, b, A, a]

$2 A B + A^2 J + B^2 J$

$-\frac{1 + A^8}{A^4}$

```
N3113 = X[i, b, h, a] X[c, j, b, i] X[k, d, j, c]
  X[p, k, a, l] X[e, o, d, p] X[m, e, l, f] X[g, m, f, n] X[o, g, n, h]
RawBracket[N3113] / J
Expand[B[N3113]]
```

X[c, j, b, i] X[e, o, d, p] X[g, m, f, n] X[i, b, h, a]
 X[k, d, j, c] X[m, e, l, f] X[o, g, n, h] X[p, k, a, l]

$8 A^7 B J^3 + 8 A B^7 J^3 + A^8 J^4 + B^8 J^4 + A^6 B^2 J^2 (25 + 3 J^2) + A^2 B^6 J^2 (25 + 3 J^2) + A^5 B^3 J (37 + 18 J^2 + J^4) + A^3 B^5 J (37 + 18 J^2 + J^4) + A^4 B^4 (25 + 37 J^2 + 8 J^4)$

$5 + \frac{1}{A^{16}} - \frac{2}{A^{12}} + \frac{3}{A^8} - \frac{4}{A^4} - 4 A^4 + 3 A^8 - 2 A^{12} + A^{16}$

```
N36 = X[a, f, b, g] X[h, c, g, b] X[d, i, c, h] X[e, l, f, m]
  X[l, e, k, d] X[m, a, n, r] X[q, o, r, n] X[p, k, o, j] X[i, p, j, q]
RawBracket[N36] / J
B[N36]
(-A^(3))^(-5) B[N36]
(* The writhe of this diagram is -5 *)
```

X[a, f, b, g] X[d, i, c, h] X[e, l, f, m] X[h, c, g, b]
 X[i, p, j, q] X[l, e, k, d] X[m, a, n, r] X[p, k, o, j] X[q, o, r, n]

$9 A^8 B J^2 + A^9 J^3 + 9 A B^8 J^5 + B^9 J^6 + 21 A^4 B^5 J^2 (5 + J^2) + 7 A^3 B^6 J^3 (11 + J^2) + A^2 B^7 J^4 (35 + J^2) + A^7 B^2 J (31 + 5 J^2) + 2 A^5 B^4 J (44 + 19 J^2) + A^6 B^3 (37 + 46 J^2 + J^4)$

$\frac{1 - 2 A^4 + 4 A^8 - 6 A^{12} + 6 A^{16} - 6 A^{20} + 5 A^{24} - 4 A^{28} + 2 A^{32} - A^{36}}{A^{21}}$

$-\frac{1 - 2 A^4 + 4 A^8 - 6 A^{12} + 6 A^{16} - 6 A^{20} + 5 A^{24} - 4 A^{28} + 2 A^{32} - A^{36}}{A^{36}}$

```
X[a, f, b, g] X[d, i, c, h] X[e, l, f, m] X[h, c, g, b]
 X[i, p, j, q] X[l, e, k, d] X[m, a, n, r] X[p, k, o, j] X[q, o, r, n]
```

$$9 A^8 B J^2 + A^9 J^3 + 9 A B^8 J^5 + B^9 J^6 + 21 A^4 B^5 J^2 (5 + J^2) + 7 A^3 B^6 J^3 (11 + J^2) + A^2 B^7 J^4 (35 + J^2) + A^7 B^2 J (31 + 5 J^2) + 2 A^5 B^4 J (44 + 19 J^2) + A^6 B^3 (37 + 46 J^2 + J^4)$$

$$\frac{1 - 2 A^4 + 4 A^8 - 6 A^{12} + 6 A^{16} - 6 A^{20} + 5 A^{24} - 4 A^{28} + 2 A^{32} - A^{36}}{A^{21}}$$

$$-\frac{1 - 2 A^4 + 4 A^8 - 6 A^{12} + 6 A^{16} - 6 A^{20} + 5 A^{24} - 4 A^{28} + 2 A^{32} - A^{36}}{A^{36}}$$

```
N42 = X[a, f, b, g] X[h, c, g, b] X[d, i, c, h] X[m, e, l, f]
  X[d, l, e, k] X[m, a, n, r] X[q, o, r, n] X[p, k, o, j] X[i, p, j, q]
RawBracket[N42] / J
B[N42]
(-A^(3))^(-1) B[N42]
```

```
X[a, f, b, g] X[d, i, c, h] X[d, l, e, k] X[h, c, g, b]
 X[i, p, j, q] X[m, a, n, r] X[m, e, l, f] X[p, k, o, j] X[q, o, r, n]
```

$$9 A^8 B J^2 + A^9 J^3 + B^9 J^4 + A B^8 J^3 (7 + 2 J^2) + A^7 B^2 J (25 + 11 J^2) + A^2 B^7 J^2 (20 + 15 J^2 + J^4) + 3 A^3 B^6 J (9 + 16 J^2 + 3 J^4) + 2 A^6 B^3 (11 + 26 J^2 + 5 J^4) + A^5 B^4 J (67 + 54 J^2 + 5 J^4) + A^4 B^5 (15 + 79 J^2 + 31 J^4 + J^6)$$

$$\frac{-1 + A^4 - A^8 + A^{12} - A^{16} + A^{20} - A^{24}}{A^9}$$

$$-\frac{-1 + A^4 - A^8 + A^{12} - A^{16} + A^{20} - A^{24}}{A^{12}}$$

```
X = A^(2) TEG + (A^(-3) - A) (-A^(4) - A^(-4)) (-A^(3))^(-2) -
  (A^(2) - A^(-2))^(2) (-A^(3))
Simplify[X]
```

$$A^3 \left(-\frac{1}{A^2} + A^2\right)^2 + \frac{\left(\frac{1}{A^3} - A\right)\left(-\frac{1}{A^4} - A^4\right)}{A^6} - \frac{\left(1 + \frac{1}{A^8} - \frac{1}{A^4} - A^4 + A^8\right)\left(-1 + A^4 + A^{12}\right)}{A^5}$$

$$\frac{-1 + A^4 - A^8 + A^{12} - A^{16} + A^{20} - A^{24}}{A^9}$$

```
(* The above is a double check in calculating N42 via using a skein calculation. *)
```

```
ML = X[i, c, j, b] X[a, i, b, h] X[g, a, h, n] X[d, p, c, o] X[d, u, e, t]
  X[l, s, k, t] X[k, q, j, p] X[v, e, u, f] X[D, f, o, g] X[C, m, D, n]
  X[w, l, v, m] X[q, z, r, A] X[z, s, y, r] X[w, C, x, B] X[A, y, B, x]
MLRAW = RawBracket[ML] / J
SUB[MLRAW]
```

```
(* This is a calculation of a link discovered by Morwen Thistlethwaite that
    has the same Jones polynomial as the unlink of two components. Here
    we calculate the bracket polynomial. The link in question has writhe
    equal to -3. Note that (-A^(3))^(3) (A^(-11)+A^(-7)) = -A^(2) - A^(-2),
 showing indeed that this bracket calculation
  sees no more than the unlink of two components. *)
```

```
X[a, i, b, h] X[A, y, B, x] X[C, m, D, n] X[d, p, c, o] X[d, u, e, t]
 X[D, f, o, g] X[g, a, h, n] X[i, c, j, b] X[k, q, j, p] X[l, s, k, t]
 X[q, z, r, A] X[v, e, u, f] X[w, C, x, B] X[w, l, v, m] X[z, s, y, r]
```

$$B^{15} J^4 + A^{15} J^5 + A B^{14} J^3 (13 + 2 J^2) + A^{14} B J^4 (11 + 4 J^2) + A^2 B^{13} J^2 (61 + 43 J^2 + J^4) + A^{13} B^2 J^3 (47 + 52 J^2 + 6 J^4) + 5 A^3 B^{12} J (25 + 55 J^2 + 11 J^4) + A^{12} B^3 J^2 (100 + 263 J^2 + 88 J^4 + 4 J^6) + A^5 B^{10} J (904 + 1618 J^2 + 460 J^4 + 21 J^6) + A^4 B^{11} (100 + 746 J^2 + 479 J^4 + 40 J^6) + A^{11} B^4 J (108 + 690 J^2 + 500 J^4 + 66 J^6 + J^8) + A^7 B^8 J (1584 + 3420 J^2 + 1317 J^4 + 113 J^6 + J^8) + A^9 B^6 J (822 + 2630 J^2 + 1394 J^4 + 157 J^6 + 2 J^8) + A^6 B^9 (340 + 2523 J^2 + 1848 J^4 + 287 J^6 + 7 J^8) + 3 A^8 B^7 (96 + 921 J^2 + 922 J^4 + 198 J^6 + 8 J^8) + A^{10} B^5 (48 + 1019 J^2 + 1497 J^4 + 418 J^6 + 21 J^8)$$

$$\frac{1}{A^{11}} + \frac{1}{A^7}$$